

Již vícekrát se mi dostal do ruky výstup se záznamy uspořádanými do bloků pod sebou v jednom sloupci a cílem byla jejich transformace do přijatelné (databázové) podoby (tj. seznam), vhodné pro filtrování a třeba jako zdroj pro kontingenční tabulku – viz obrázek. Pracovně tomuto typu úlohy říkám „věšení prádla“.



Rozřazení dat do sloupců

Bloky nejsou stejně dlouhé (vysoké), tj. záznamy nemají vždy vyplněné všechny položky. Pro zpracování potřebujeme znát maximální počet položek v bloku (budoucích sloupců, tj. polí) a počáteční položku nového bloku. Nabízím následující způsob řešení:

```

1  Option Base 1
2
3  Const strOddelovacPoleHodnota As String = ":"
4
5  Sub TransformaceDat()
6
7      Dim rngData As Range
8      Dim rngBunka As Range
9
10     Dim aPole
11     Dim astrPolozky
12     Dim strPrvniPolozka As String
13
14     'definice polí
15     astrPolozky = Array("Student", "Datum", "Předmět 1", "Předmět 2", _
16         "Předmět 3", "Předmět 4", "Předmět 5")
17
18     'referenční pole (začíná jím blok)
19     strPrvniPolozka = astrPolozky(1)
20
21     'zdrojová data
22     Set rngData = Range(Range("A1"), Range("A1").End(xlDown))
23
24     'počet sloupců
25     iPocetSloupcu = UBound(astrPolozky)
26     'počet bloků (výskytů referenční položky)
27     iPocetBloku = WorksheetFunction.CountIf(rngData, "=" & strPrvniPolozka & "**")
28
29     ReDim aPole(1 To iPocetBloku, 1 To iPocetSloupcu)
30
31     'pro každou buňku
32     For Each rngBunka In rngData
33
34         'rozdělení buňky na název pole a jeho hodnotu
35         aPoleBunka = Split(rngBunka.Text, strOddelovacPoleHodnota)
36
37         'pozice nalezeného pole
38         'ořez od mezer před a za textem
39         iPole = Application.Match(WorksheetFunction.Trim(aPoleBunka(0)), _
40             astrPolozky, 0)
41
42         If iPole = 1 Then
43             'zjištěno pole Student ... nový řádek
44             iZaznam = iZaznam + 1
45         End If
46
47         'ořez od mezer před a za textem
48         aPole(iZaznam, iPole) = WorksheetFunction.Trim(aPoleBunka(1))
49
50     Next rngBunka
51
52     'naplnění hlavičky
53     Range("C1").Resize(1, iPocetSloupcu) = astrPolozky
54
55     'naplnění oblasti buněk
56     Range("C2").Resize(iPocetBloku, iPocetSloupcu) = aPole
57
58 End Sub

```

Možná vás napadlo použít Data / Text do sloupců k rozdělení názvu položky od její hodnoty na základě dvojtečky. Ano, je to možné, ovšem větší smysl by to mělo pouze v případě, že byste chtěli poté první sloupec filtrovat na jedinečné položky a zjišťovat onen maximální počet typů položek v bloku (záznamu). Otázkou je, jestli daný vzorek dat obsahuje všechny existující. Já jsem pro to raději zjistit daný údaj od zadavatele a v kódu s ním již napevno počítat (viz astrPolozky).

Příklad ke stažení:

[excel-rozradit.zip](#)